

Wprowadzenie do programu gnuplot

Bogdan Kreczmer

ZPCiR ICT PWR

pokój 307 budynek C3

kreczmer@ict.pwr.wroc.pl

Copyright ©2003 Bogdan Kreczmer*

* Niniejszy dokument zawiera materiały do wykładu na temat wizualizacji danych sensorycznych. Jest on udostępniony pod warunkiem wykorzystania wyłącznie do własnych prywatnych potrzeb i może on być kopiowany wyłącznie w całości, razem z niniejszą stroną tytułową.

gnuplot

Główni autorzy: Thomas Williams, Colin Kelley

<http://www.gnuplot.info/>

<ftp://ftp.gnuplot.info/pub/gnuplot/>

gnuplot jest programem przeznaczonym do:

- tworzenia rysunków wykresów funkcji jedno i dwuargumentowych (funkcje mogą być parametryzowane),
- obrazowania danych pomiarowych,
- tworzenia wykresów interpolujących przebiegi funkcji na podstawie zbioru danych pomiarowych. Przy interpolacji brane są pod uwagę błędy związane z zadanymi wartościami.

Jest to program zorientowany na polecenia tekstowe w pracy interaktywnej lub wsadowej.

Tryby pracy

*Możliwe sposoby pracy z programem **gnuplot**:*

Tryb interaktywny

```
SunOS 5> gnuplot
gnuplot>_
      . . .
gnuplot>exit
SunOS 5>_
```

Tryb potokowy

```
SunOS 5> generator_polecen | gnuplot
```

Tryb wsadowy

```
SunOS 5> gnuplot plik_komend.asci
```

Podpowiedzi - help

```
gnuplot> help
```

'gnuplot' is a command-driven interactive function and data plotting program. It is case sensitive (commands and function names written in lowercase are not the same as those written in CAPS). All command names may be abbreviated as long as the abbreviation is not ambiguous. Any number of commands may

. . .

Help topics available:

batch/interactive	bugs	commands	comments
coordinates	copyright	environment	expressions
glossary	graphical	introduction	line-editing
new-features	old_bugs	plotting	seeking-assistance
set	show	startup	substitution
syntax	time/date		

Help topic: _

Najprostszy wykres

Praca interakcyjna:


```
gnuplot>plot sin(x)
```

Przetwarzanie wsadowe:

```
SunOS 5>gnuplot sin.cmd
```

Plik "sin.cmd":

```
#  
# Rysowanie  
#  
plot sin(x)  
#  
# Zatrzymanie aż do  
# naciśnięcia klawisza  
# ENTER  
#  
pause -1
```



Rysunek 1: Wykres funkcji sinus.

Najprostszy wykres


`plot sin(x) with lines 2`

`plot sin(x) w l 2`


**Składnia sekcji `with`
dla polecenia `plot`:**

```
with <style> { {linestyle | ls <line_style>}  
              | {{linetype | lt <line_type>} {linewidth | lw <line_width>}  
              {pointtype | pt <point_type>} {pointsize | ps <point_size>}} }
```

```
<style> = lines | points | linespoints | impulses | dots | steps | fsteps |  
          histeps | errorbars | xerrorbars | yerrorbars | xyerrorbars |  
          boxes | xyerrorbars | boxes | boxerrorbars | boxxyerrorbars |  
          financebars | candlesticks | candlesticks | vector
```



Polecenie test




gnuplot>show terminal

terminal type is x11


gnuplot>test

Rysunek 2: Test "terminala".


Przykłady stylów rysowania




plot sin(x) with 1 lt 2




plot sin(x) with 1 lt 3 lw 3




plot sin(x) with 1 lt 7 lw 9




plot sin(x) with points




plot sin(x) with points pt 1 ps 4




plot sin(x) with p lt 2 pt 2 ps 8



plot sin(x) with linespoints lt 2




... w linesp lt 3 lw 3 pt 1 ps 4




... w linesp lt 7 lw 9 pt 2 ps 8


Przykłady stylów rysowania




plot sin(x) with impulses




plot sin(x) with i lt 2 lw 2




plot sin(x) with i lt 3 lw 8




plot sin(x) with steps




plot sin(x) w fsteps lt 2 lw 2




plot sin(x) w histeps lt 3 lw 9



plot sin(x) with boxes



... w candlesticks




... w candlesticks lt 1 lw 3

Oś zera

Polecenia:

```
gnuplot> set yzeroaxis  
gnuplot> set xzeroaxis lt 1 lw 3  
gnuplot> plot sin(x)
```




Rysunek 3: Wykres funkcji sinus.

Z radianów na stopnie

Polecenia:


```
gnuplot> set angles degrees  
gnuplot> plot [0:360] sin(x)
```



Rysunek 4: Wykres funkcji sinus.

Funkcje parametryczne

```
set parametric  
plot sin(t),cos(t)
```




Rysunek 5: Wykres funkcji parametrycznej.


Wzajemna proporcjonalność osi

```
gnuplot> . . .  
gnuplot> set size ratio 1  
gnuplot> set parametric  
gnuplot> plot sin(t),cos(t)
```


noratio



ratio: 1



ratio: 1.33




`set size {{no}square | ratio <r> | noratio} {<xscale>,<yscale>}`

Dane z pliku

```
gnuplot>plot ''dane.asc''
```

Plik "dane.asc" :

```
#  
# Przykład danych  
# liczbowych  
#  x    y  
#  
2    3  
3    7  
4    2  
4.4  1.2  
5    1  
7    2  
8    6  
10   9
```




Rysunek 6: Wykres zadany poprzez zbior punktów.

Wczytywanie z wejścia standardowego

```
gnuplot> plot "< awk '/^ *[^#]/ { print $2, $1}' dane2.asc" with lines
```

Plik "dane2.asc":

```
#  
# Przykład danych  
# liczbowych  
#  x    y    Dy  
#  
#  2    3    0.9  
#  3    7    0.5  
#  . . .
```




Kolejność współrzędnych - using

```
gnuplot> plot 'dane.asc' using 2:1 with lines
```

Plik "dane.asc" :

```
#  
# Przykład danych  
# liczbowych  
#  y    x  
#  
2    3  
3    7  
4    2  
4.4  1.2  
5    1  
7    2  
8    6  
10   9
```




Rysunek 7: Wykres zadany poprzez zbior punktów z jawnym wskazaniem współrzędnej x i y .

Rysowanie wielkości błędu - errorbars

```
gnuplot> plot "dane2.asc" with errorbars
```

Plik "dane2.asc":

```
#  
# Przykład danych  
# liczbowych  
#  x    y    Dy  
#  
2    3    0.9  
3    7    0.5  
4    2    0.8  
4.4  1.2  0.7  
5    1    0.2  
7    2    0.6  
8    6    0.3  
10   9    0.6
```



Rysunek 8: Wykres zadany poprzez zbior punktów.


Dane razem z poleceniami

plot '-' with lines, '-' notitle with errorbars

```
#
# x   y   Dy
#
2    3    0.9
3    7    0.5
4    2    0.8
4.4  1.2  0.7
5    1    0.2
7    2    0.6
8    6    0.3
10   9    0.6
```

e

e




Rysowanie wielkości błędu - errorbars

```
gnuplot> plot "dane2.asc" with lines, "dane2.asc" with errorbars
```

Plik "dane2.asc":

```
#  
# Przykład danych  
# liczbowych  
#  x    y    Dy  
#  
2    3    0.9  
3    7    0.5  
4    2    0.8  
4.4  1.2  0.7  
5    1    0.2  
7    2    0.6  
8    6    0.3  
10   9    0.6
```




Rysunek 9: Wykres zadany poprzez zbior punktów.

Rysowanie wielkości błędu - xyerrorbars

```
gnuplot> plot "dane3.asc" with xyerrorbars
```

Plik "dane3.asc":

```
#  
# Przykład danych  
# liczbowych  
# x    y    Dx    Dy  
#  
2     3     0.5   0.9  
3     7     0.2   0.5  
4     2     0.9   0.8  
4.4   1.2   0.4   0.7  
5     1     0.6   0.2  
7     2     0.8   0.6  
8     6     0.4   0.3  
10    9     0.4   0.6
```




Rysunek 10: Wykres zadany poprzez zbior punktów z informacją o błędach.

Rysowanie z wygładzaniem

```
gnuplot> plot "dane.asc" smooth csplines
```

Plik "dane.asc" :

```
#  
# Przykład danych  
# liczbowych  
#  x    y  
#  
2    3  
3    7  
4    2  
4.4  1.2  
5    1  
7    2  
8    6  
10   9
```




Rysunek 11: Wykres zadany poprzez zbior punktów. Rysunek z wygładzaniem.

Rysowanie z wygładzaniem

```
gnuplot> plot 'dane.asc' smooth csplines w l lw 3,\n           'dane.asc' smooth bezier w l lt 3 lw 2, "dane.asc" w linesp lt 2 ps 3
```

Plik "dane.asc":

```
#  
# Przykład danych  
# liczbowych  
# x y  
#  
2 3  
3 7  
4 2  
4.4 1.2  
5 1  
7 2  
8 6  
10 9
```




Rysunek 12: Wykresy zadane tym samym zbiorem punktów.

Wykładanie parametryzowane

```
gnuplot>plot 'dane2.asc' using 1:2:(70) smooth acsplines title "Współczynnik 70",\  
            'dane2.asc' using 1:2:(1) smooth acsplines title "Współczynnik 1",\  
            'dane2.asc' notitle with errorbars
```

Plik "dane2.asc":

```
#  
# Przykład danych  
# liczbowych  
# x    y    Dy  
#  
2     3     0.9  
3     7     0.5  
4     2     0.8  
4.4   1.2   0.7  
5     1     0.2  
7     2     0.6  
8     6     0.3  
10    9     0.6
```




Rysunek 13: Wykładanie parametryzowane.

Wygładzanie parametryzowane

```
gnuplot>plot . . . \
'dane2.asc' using 1:2:(40/($3*$3)) smooth acsplines\
title "Wspolczynnik wyliczany" with lines lw 3
```

Plik "dane2.asc":

```
#
# Przykład danych
# liczbowych
# x    y    Dy
#
  2    3    0.9
  3    7    0.5
  4    2    0.8
  4.4  1.2  0.7
  5    1    0.2
  7    2    0.6
  8    6    0.3
 10    9    0.6
```




Rysunek 14: Wygładzanie parametryzowane wartościami z trzeciej kolumny.

Zakres - range

```
gnuplot> plot [1:11] [0:10] 'dane.asc' with lines
```


```
#  
# Drugi sposób:  
#  
set xrange [1:11]  
set yrange [0:10]  
plot 'dane.asc' w lines
```



Rysunek 15: Wykres zadany poprzez zbior punktów z własnym ustawieniem zakresu zmian współrzędnych.

Nazwa wykresu - title

```
gnuplot> plot [1:11] [0:10] 'dane.asc' title "Przykładowe dane" with lines\  
          'dane.asc' smooth bezier notitle w lines
```




Rysunek 16: Wykres zadany poprzez zbior punktów z własnym ustawieniem zakresu zmian współrzędnych.

Definiowanie własnych funkcji

```
gnuplot> ogrD(x,d)=x-d
gnuplot> ogrG(x,d)=x+d
gnuplot> wspol(x,s)=s/(x*x)
gnuplot> plot 'dane2.asc' using 1:(ogrG($2,$3)) title "Ograniczenie gorne" with lines,\
'dane2.asc' using 1:(ogrD($2,$3)) title "Ograniczenie dolne" with lines,\
'dane2.asc' notitle with errorbars,\
'dane2.asc' using 1:2:(wspol($3,40)) smooth acsplines\
title "Wspolczynnik wyliczany" with lines lw 3
```

Plik "dane2.asc":

```
#
# Przyklad danych
# liczbowych
#  x    y    Dy
#
#  2     3    0.9
#  3     7    0.5
#  4     2    0.8
#  4.4 1.2    0.7
#  . . .
```




Parametryzacja wykresu

```
gnuplot> . . .
gnuplot> moje_x(x,y)=0.2*($2)*sin($1*3.14/9)
gnuplot> moje_y(x,y)=0.2*($2)*cos($1*3.14/9)
gnuplot> wspol(x,s)=s/(x*x)
gnuplot> set parametric
gnuplot> plot 'dane2.asc' using (moje_x($1,$2)):(moje_y($1,$2))\
        title "Krzywa modulowana" with lines,\
        'dane2.asc' using (moje_x($1,$2)):(moje_y($1,$2)):(wspol($3,40))\
        smooth acsplines title "Wygladzony",\
        sin(t),cos(t)
```

Plik "dane2.asc":

```
#
# Przyklad danych
# liczbowych
#  x    y    Dy
#
#      2    3    0.9
#      3    7    0.5
#      . . .
```




Biegunowy układ współrzędnych

```
gnuplot> . . .
gnuplot> set grid polar
gnuplot> set xtics axis
gnuplot> set ytics axis
gnuplot> set polar
gnuplot> set size ratio 1
gnuplot> plot 'dane2.asc' using ($1/5):($2/5)\
               title "Krzywa modulowana" with lines,\
               'dane2.asc' using ($1/5):($2/5):(wspol($3,40))\
               smooth acsplines title "Wygladzony",\
               sin(t), sin(t)+1.5
```

Plik "dane2.asc":


```
#
# Przykład danych
# liczbowych
#  x    y    Dy
#
#  2    3    0.9
#  3    7    0.5
#  . . .
```




Siatka i podziałka

```
gnuplot> set y2tics
gnuplot> set my2tics 2
gnuplot> set mxtics
gnuplot> set grid
gnuplot> plot 'dane2.asc' with lines
```


set grid



set grid xtics



set grid xtics ytics mytics



```
set grid {{no}{m}xtics} {{no}{m}ytics} {{no}{m}ztics}
        {{no}{m}x2tics} {{no}{m}y2tics}
        {polar {<angle>}} { {linestyle <major_linestyle>}
                             | {linetype | lt <major_linetype>} ...
```